
1. Machine Description
In the packaging industry, one step in the process of getting

a product to market is the labeling of that product. This Motion-
Code example focuses on a label applicator machine (Figure 1)
that applies a product brand label to a cardboard box. The label
covers the entire length of the box, so it must be aligned with
the leading edge of the box.

The process begins with boxes being fed at a constant speed
into the machine on a conveyor. The boxes are pushed along the
conveyor by mechanical catches that keep the boxes a fixed dis-
tance apart.

The label mechanism consists of a drive wheel, an applicator
wheel, and a spool. The drive wheel intermittently pulls the label
web from the spool as the web passes over the applicator wheel,
which forces the label onto the box. An open loop offset is appli-
ed to the spool to maintain web tension.

Because the labels are close together on the web, the web
must start and stop. The label is applied by matching speed be-
tween the box and applicator wheel. When the conveyor reaches
a specified position, the label drive wheel accelerates to this
matched speed, applies the label, and then decelerates to a stop.

Because the web can slip, it has registration marks that are
used to insure proper placement of each label. A sensor reads the
position of the registration mark, and after the web decelerates,
the drive wheel repositions to account for any errors in web
position.

Figure 1. Label Applicator Machine

2. Requirements
This section summarizes the control system requirements

for the label applicator machine shown in Figure 1.

(1) Three axes: Conveyer (A), Label Drive Wheel (B),
and Label Spool (C)

(2) Home input for homing the conveyer
(3) Ramped gearing of label drive wheel to conveyor
(4) High speed position latch digital input to read 

registration positions
(5) Position correction of label drive wheel on-the-fly

to account for registration
(6) No user interface is required

3. Components Selected
This section describes the Galil hardware and software

products chosen to implement the machine’s control sys-
tem. Below is a complete bill of materials followed by a
description of major components.

Part number Description Unit Price (U.S)
Qty 1/Qty 100

DMC-2133 3-axis Ethernet, RS232 controller card with 96-pin DIN connector $1045/$725

-DIN DIN-rail mount option $100/$50

-DC24 DC-to-DC converter option for 18V to 36V $100/$70

AMP-20540 Amplifier Board with four 500 W PWM drives for brush or brushless servos $795/ $495

BLM-N23-50-1000 or equivalent NEMA 23 Brushless Servo Motor, 1000 ppr encoder with Hall Sensors Consult manufacturer x3

CPS-12-24 or equivalent 24 V, 200 Watt Power Supply Consult manufacturer

WSDK Servo Tuning Software Servo Tuning and Analysis Software $195 (one time)

Table 1.  Bill of Materials for Label Applicator Control System



Controller: DMC-2133-DC24-DIN

Since the label applicator does not require a host PC,
we choose the DMC-2133 stand-alone motion controller.
A laptop may be connected via Ethernet or RS-232 to per-
form servo tuning with the WSDK and to load the appli-
cation program.

Figure 2. DMC-21x3. 1 to 4 axes (top) and 
5 to 8 axes (bottom)

Motor: BLM-N23-50-1000

For maintenance-free operation, we choose brushless
motors. Galil’s NEMA 23 #BLM-N23-50-1000 brushless
motors, or equivalent, are appropriate because all axes re-
quire less than 0.3 Nm of continuous torque. Increment-
al encoders with 1000 cycles per revolution are installed
on the motors resulting in 4000 quadrature counts per re-
volution. Hall sensors are not required on the motors as
the incremental encoders provide commutation tracks for
input to the amplifiers.

Amplifier: AMP-20540

To drive the motors, we choose the very compact AMP-
20540, which is a four-axis brush/brushless amplifier cap-
able of 500W per axis. This unit connects directly to the
DMC-2133, eliminating all the wiring which typically
connects a controller
to the amplifiers.

Figure 3. AMP-20540 four-axis brushless servo
amplifier (500W per axis)

4. Implementation
This section details how the components selected above

were used to implement the control system.

Gearing (GA, GR, GD)

The Galil gearing function is used to set a ratio between
the conveyor (master) and label drive wheel (slave). The
GA command sets which axes are masters and slaves, and
GR sets the gear ratio. The GD command is used to set the
distance of the master axis over which the slave ratio is
ramped.

GAB = A ;'Set conveyor as master axis for gearing
GDB = 800 ;'Label drive wheel ramps over 800 conveyer counts
GRB = 0.75 ;'Set label drive wheel gear ratio to begin accel

High Speed Position Latch (AL, _AL, _RL)

Galil motion controllers feature a dedicated digital
input called the high-speed position latch, which is an
edge-based hardware trigger. These dedicated inputs store
the corresponding axis position within 0.1µs of the input
transition. The latch input is used in this application to
accurately maintain the registration position of the labels.

After arming the latch with the ALB command and
upon perceiving a state transition on the input, the con-
troller will automatically place the encoder position in
a register, which can be read with _RLB. _ALB returns
a Boolean value indicating whether the latch has triggered.
The CN command controls whether the position is latched
on a rising (CN,,1) or falling (CN,,-1) edge.

Calculations

The first step in writing the code for this application
is to calculate the required speeds and distances in counts.
Figure 4 shows the machine geometry. The boxes move
at v = 10 in/s and are 10” long with 6” in between each
box. The labels are 10” long with 2” in between each
label. The registration marks are centered between each
label on the web (1” from each label edge).  

Since the box is moving at a constant velocity v, it
moves twice as far (xbox = vt) in a given time t as the web,
which must accelerate to speed (xweb = 1/2 at2 = 1/2vt).
Therefore, the home sensor is placed 2” (2 * 1”) from the
applicator wheel, which gives the label a 1” head start. 
It takes  t = 2 xweb/v = (2 * 1 in) / (10 in/s) = 0.2 s for
the web to accelerate to speed and (10 in/s) / (10 in) = 1 s
to apply the label. With these numbers, graphs of position
versus time and velocity versus time may be calculated
(Figure 5). These graphs assume the conveyor has already
reached its steady state velocity of 10 inches/second.

Figure 4. Label applicator machine geometry at the instant the web begins to accelerate

Figure 5. Position and speed versus time of the box and label web (the process repeats every 1.6 seconds)

Given the wheel diameters and encoder resolutions from Figure 4, Table 2 calculates the linear resolutions in counts
per inch and the maximum speeds in counts per second for both the conveyer and label drive wheel. From the table, the
electronic gear ratio GR for matched speed is:

3000 counts/s (drive wheel) : 4000 counts/s (conveyor) = 0.75 :1

With the linear resolution (counts/in) of the conveyer, table 3 translates key positions from Figure 5 into counts.

Program Description

1. The code starts automatically upon power up with the #AUTO routine.

2. The controller homes the conveyor, which places the first box 800 counts ahead of the label applicator wheel.  
This position is defined as zero.

3. The label drive axis is homed to the first label registration mark.

4. The conveyor begins its rotation, transporting the boxes to the label applicator wheel. Given that the boxes occur 
every 16 inches (10 inch box + 6 inch space), the code monitors for conveyor positions of 6400, 12800, 19200, etc.

5. When a box is in position for label application, the label drive wheel accelerates. The acceleration takes place 
over a distance of 800 counts of the conveyor.

6. Once the label is at matched speed with the box, the code waits for a total of 12” (4800 counts) to pass, allowing 
the label to be applied.

7. After the label has been applied, the label drive wheel is disengaged over a conveyer distance of 800 counts.

8. After the gearing is disengaged, the registration correction is made.

9. If the registration mark has been seen, an incremental move is made to realign the label drive wheel to the
registration mark.

10. If the mark has not been seen, the label drive wheel is moved until the latch is triggered. Once triggered,
an incremental move is made to re-align the label drive wheel to the registration mark.

11. This process is repeated

Axis Given Circumference Given Linear Required Speed
Diameter (in/rev) Encoder Resolution Speed (counts/s)
(in) Resolution (counts/in) (in/s)

(counts/rev)

Conveyer 6.37 20.0 8000 400 10 4000

Label Drive Wheel 4.24 13.3 4000 300 10 3000

Table 2. Calculations for linear resolution in counts and speed in counts per second

Event Time (s) Conveyer Conveyor
Position Position
(in) (counts) 

Web begins accelerating (Figure 4) 0 0 0

Web at speed. Begin applying label 0.2 2 800

Finished applying label. Begin decelerating web. 1.2 12 4800

Web done decelerating. Make registration correction. 1.4 14 5600

Web is at rest 1.6 16 6400

Table 3. Calculation of key conveyer positions in counts



#AUTO ;'Runs on power up
'Set motion parameters for homing
AC*=500000 ;'set acceleration
DC*=500000 ;'set deceleration

'Apply open-loop offset to label spool
KDC = 0 ;'set derivative gain to 0
KPC = 0 ;'set proportional gain to 0
KIC = 0 ;'set proportional gain to 0
OFC = -0.5 ;'Set negative offset to apply tension to web.

'Home conveyor to home sensor at high speed
SPA=100000 ;'set speed
FEA ;'find the home switch
BGA ;'begin conveyer motion
AMA ;'wait until home switch is found

'Come off home switch slowly
SPA = 500 ;'Set low conveyor speed
FEA ;'Home conveyor at low speed for accuracy
BGA ;'begin conveyer motion
AMA ;'wait until we come off home switch
DPA = 0 ;'Define conveyor position as 0

'Home label drive wheel to registration mark on label web
ALB ;'Arm latch to watch for first registration mark
JGB=250 ;'Jog label drive wheel towards registration mark
BGB ;'Begin label drive wheel motion
#Wait; JP#Wait, _ALB = 1 ;'Wait until registration mark is seen
STB ;'Stop label drive wheel
AMB ;'Wait until label drive wheel motion is complete
PAB = _RLB ;'Move label drive wheel back to reg mark position
BGB ;'Begin label drive wheel motion
AMB ;'Wait until label drive wheel motion is complete
DPB=0 ;'Define that position as 0

'initialize gearing
GAB = A ;'Set conveyor as master axis for gearing
GDB = 800 ;'Label drive wheel ramps over 800 conveyer counts

'start the conveyer
JGA = 4000 ;'Set jog speed of conveyor axis
BGA ;'Begin jogging conveyor

BoxPos = 0 ;'Leading box positions:  0, 6400, 12800...

#Apply ;'Loops once per label
ALB ;'Arm registration latch

IF _TPX >= BoxPos ;'If conveyer position indicates a new box
GRB = 0.75 ;'Set label drive wheel gear ratio to begin accel
MF BoxPos + 4800 ;'Wait until label is applied (2" + 10")
GRB = 0 ;'Begin gearing decel of label drive wheel
MF BoxPos + 5600 ;'Wait for decel of drive wheel complete (14")
BoxPos = BoxPos + 6400 ;'Increment box modulus position (16")
Correct = 0 ;'Whether the registration sensor has tripped

Program listing

The complete program used for this article is included here.  The comments explain the code in detail.

(Continued next page)



#Correct ;'Registration correction loop
SPB=100000 ;'set correction speed

IF _ALB = 0 ;'If registration sensor has triggered
PAB = _RLB ;'Move back to the latched position
BGB ;'Begin label drive wheel motion
AMB ;'Wait until label drive wheel motion is complete
Correct = 1 ;'Indicate that we may exit the #Correct loop

ELSE ;'If registration sensor has not triggered
IPB = 10 ;'Increment label drive wheel 10 counts
AMB ;'Wait until incremental motion is complete

ENDIF
JP#Correct, Correct = 0 ;'Continue if registration sensor hasn’t triggered

ENDIF

JP#Apply ;'Next label

EN

